[bookmark: _Toc60054457]Supplementary Text 1. Code for Sales Prediction Model 
# Code were run by OpenBUGS software 
#Description and example code for proper Conditional Autoregressive (CAR) Model are also available in GeoBugs manual. https://www.mrc-bsu.cam.ac.uk/wp-content/uploads/geobugs12manual.pdf
#OpenBUGS code to generate sales of plain yogurt (same code can be applied to the other food categories)
# YOGPL stands for plain yogurt.
# Note that Y_YOGPL correspond to Q_ij in the main text and contain observed and missing sales   
model{
  ### Non-sugar yogurt sales #########################################################################  
  for(j in 1:J_YOGPL){    #J_YOGPL inicdates number of stores 
    # log sales of soda as normal
    logY_YOGPL[j] ~ dnorm(mu.y_YOGPL[j], tau.y_YOGPL)    
    # Mean of log sales specified by the sum of chain random effect and area-level effect for area 
    # ID_chain_YOGPL is an index for chain where store j belongs to 
    # ID_neigh_YOGPL is an index for area where store j belongs to 
    mu.y_YOGPL[j] <- betaChain_YOGPL[ID_chain_YOGPL[j]]  + Z_YOGPL[ID_neigh_YOGPL[j]]
    # generate non-log soda sales 
    Y_YOGPL[j] <- exp(logY_YOGPL[j])
  } 

  # Area-level effects Z containing area-level covariates and spatial random effect S_YOGPL
  # N_neigh is number of areas (n=193)
  for(i in 1:N_neigh) {
    Z_YOGPL[i] <- 
      betaEduc_YOGPL*educ_neigh[i]  + betaIncome_YOGPL*income_neigh[i]+ betaPopDensity_YOGPL*popDensity_neigh[i] + 
      betaFamSize_YOGPL*famSize_neigh[i] + betaYoung_YOGPL*young_neigh[i] + betaEmpRate_YOGPL*empRate_neigh[i]  +   S_YOGPL[i]  
  }

  # Specification of proper CAR prior for spatially structured area-level random effect. See the link above for details 
  S_YOGPL[1:N_neigh] ~ car.proper(mu[], C[], adj_neigh[], num_neigh[], M[], tau.car_YOGPL, gamma)

  for(i in 1:N_neigh){
    mu[i]<-betaZeroArea_YOGPL
    M[i] <- 1/num_neigh[i]
  }  
  
  cumsum[1] <- 0
  for(i in 2:(N_neigh+1)) {  cumsum[i] <- sum(num_neigh[1:(i-1)])   } 
  for(k in 1 : sumNumNeigh) { 
         for(i in 1:N_neigh) {    pick[k,i] <- step(k - cumsum[i] - epsilon)  * step(cumsum[i+1] - k)   
    }                                                       
    C[k] <- 1 / inprod(num_neigh[], pick[k,])   
  }
  epsilon <- 0.0001
  gamma.min <- min.bound(C[], adj_neigh[], num_neigh[], M[])
  gamma.max <- max.bound(C[], adj_neigh[], num_neigh[], M[])
  gamma ~ dunif(gamma.min, gamma.max)
  
  # Store chain random effect 
  for(k in 1:J_YOGPL_chain){    
    betaChain_YOGPL[k] ~dnorm(betaZero_YOGPL, tau.chain_YOGPL)
  }  

  # Standard deviations
  sigma.chain_YOGPL <-1/sqrt(tau.chain_YOGPL)
  sigma.car_YOGPL <- sqrt(1/tau.car_YOGPL)
  sigma.y_YOGPL <- sqrt(1/tau.y_YOGPL)

  #Prior probabilities 
  tau.y_YOGPL ~ dgamma(0.5, 0.01)
  tau.car_YOGPL ~ dgamma(0.5, 0.01)
  tau.chain_YOGPL ~ dgamma(0.5, 0.01)
  
  betaZero_YOGPL ~ dnorm(0,0.01)
  betaZeroArea_YOGPL <- 0
  betaEduc_YOGPL ~ dnorm(0,0.01)
  betaIncome_YOGPL ~ dnorm(0,0.01)
  betaPopDensity_YOGPL ~ dnorm(0,0.01)
  betaFamSize_YOGPL  ~ dnorm(0,0.01)
  betaYoung_YOGPL ~ dnorm(0,0.01)
  betaEmpRate_YOGPL ~ dnorm(0,0.01)

# End of Sales prediction model for yogurt 

### Extra step- indicator generation 
# Multiply the column vector of yogurt (plain) sales with matrix whose elements are a pairwise area-level store visit probability, between store and area multiplied with population identity (matrix G* in the main text)
# N_Neigh is the number of area (neighborhoods),, 193
# n_Neigh is an iterator (index) for area i
# odMX_ YOGPL is matrix G*  
# Y_ YOGPL is exponentiated sales quantity of stores selling yogurt 
for(n_Neigh in 1:N_Neigh){
    #Get linear combination of all area’s sales multiplied with the OD weights for eatch area N_Neigh
    # Note that OpenBugs does not allow matrix multiplication, thus a series of inprod() – inner products - is needed. 
    neighSales_combined_ YOGPL[n_Neigh] <- inprod(Y_YOGPL[1:J_OD_YOGPL], odMX_ YOGPL[,n_Neigh]) 
   
    # finally divide the value (purchased quantity in an area n_neigh by population in each area to make the quantity purchased per resident 
    # indicator_ YOGPL corresponds to X in the main text
    indicator_ YOGPL[n_Neigh] <- neighSales_combined_ YOGPL[n_Neigh]/ pop_Neigh[n_Neigh]
} 


#End of script 
} 

